CANADA

DEPARTMENT OF
ENERGY, MINES AND RESOURCES
MINES BRANCH
OTTAWA

COMPUTER PROGRAMS FOR X-RAY CRYSTALLOGRAPHY PART I: BOND AND ANGLE SCAN PROGRAM

E. J. GABE

MINERAL SCIENCES DIVISION

MARCH 1967

1867 1967

Research Report R 189

Price 50 cents

Available by mail from the Queen's Printer, Ottawa,
and at the following Canadian Government bookshops:

OTTAWA
Daly Building, Corner Mackenzie and Rideau

TORONTO
221 Yonge Street

MONTREAL
Æterna-Vie Building, 1182 St. Catherine St. West

WINNIPEG
Mall Center Building, 499 Portage Avenue

VANCOUVER
657 Granville Avenue

HALIFAX
1737 Barrington Street

or through your bookseller

A deposit copy of this publication is also available
for reference in public libraries across Canada

Price 50 cents Catalogue No. M38-1/189

Price subject to change without notice

ROGER DUHAMEL, F.R.S.C.
Queen's Printer and Controller of Stationery
Ottawa, Canada
1967

Mines Branch Research Report R 189

COMPUTER PROGRAMS FOR X-RAY CRYSTALLOGRAPHY.

PART I:

BOND AND ANGLE SCAN PROGRAM

by

E. J. Gabe*

- - -

ABSTRACT

This program, which is the first in the series, is designed to facilitate the interpretation of the results of crystal structure analysis. Using the unit cell data, the fractional atomic coordinates and the symmetry operations, the program finds the coordination, in terms of distances and angles, around a specified group of atoms of interest. The program will handle all types of symmetry and there are essentially no limits on the size of structure that can be dealt with, other than the size of the memory of the computer.

*Research Scientist, Mineralogy Section, Mineral Sciences Division, Mines Branch, Department of Energy, Mines and Resources, Ottawa, Canada.

Direction des mines

Rapport de recherches R189

PROGRAMMES D'ORDINATEUR POUR LA RADIOCRISTALLOGRAPHIE.

PARTIE I:

PROGRAMME D'EXPLORATION DES LIENS ET DES ANGLES

par

E. J. Gabe*

- - -

RÉSUMÉ

Ce programme, le premier d'une série, est conçu pour faciliter l'interprétation des résultats de l'analyse des structures cristallines. A l'aide des données de l'unité de la maille, des coordonnées fractionnaires atomiques et des opérations de symétrie, le programme permet de trouver, selon les distances et les angles, la coordination qui existe dans un groupe donné d'atomes. Le programme permettra d'étudier tous les genres de symétrie et il n'y a à peu près aucune limite à la dimension de la structure étudiée, excepté celle de la mémoire de l'ordinateur.

*Préposé à la recherche, Section de la minéralogie, Division des sciences minérales, Direction des mines, ministère de l'Énergie, des Mines et des Ressources, Ottawa, Canada.

CONTENTS

- - - - -

INTRODUCTION

A series of computer programs to deal with all aspects of computing in the field of structural X-ray crystallography is being initiated at the Mines Branch. This report describes the first program of the series, and new programs will be added as they are written. The system is being designed for use on an IBM/360 model 65 computer with FORTRAN IVH or IVG. It will deal with intensity data collection and reduction, structure solution and refinement, and interpretation and evaluation of the results.

The function of the program described in this report is to find all interatomic distances, less than a variable limit, surrounding a stated group of atoms and then to calculate all angles between the distances found.

GENERAL DESCRIPTION

In order to study the coordination of a group of atoms, it is necessary to generate all the symmetry-equivalent atoms in the surrounding cells and then to find all distances, less than a suitable limit, between the "original" atoms and the generated ones. In the present program a group of original atoms--the asymmetric unit-- are used as input, and these, together with symmetry matrices, are used to generate the surrounding atoms. In order to save computing time, a distance related to the maximum dimensions of the

original group and the bond limit is first calculated and then used as the radius of a large sphere about the original group, into which all the symmetry equivalent atoms must fall. All equivalents falling within this sphere are saved, and then this list is systematically searched or scanned for all distances between the original and saved atoms less than the limit. For each "original" atom, a list of distances is kept and all angles between these distances are calculated. The output consists of lists of angles and bonds centred on each original atom in turn.

The program is set up to accept up to 99 original atoms, 96 equivalents of any one atom, 2,500 saved symmetry equivalents, and up to 400 bond lengths to any one atom. These limitations are regarded as giving adequate scope for most computations, and are easily changed. Several scans may be run at one time, and the computation time for the scans of typical sulphide mineral structures is less than one minute.

DETAILS OF CALCULATION

During the calculation, all coordinates are orthogonalized by the equation:

$$\underset{\sim}{X_j} = \underset{\sim}{T_{ij}}\ \underset{\sim}{x_i}$$

where X_i = orthogonal coordinates in Å;

x_j = fractional coordinates of original unit cell;

and T_{ij} = orthogonality matrix, as follows:

$$T_{ij} = \begin{bmatrix} a & b\cos\gamma & c\cos\beta \\ 0 & b\sin\gamma & \dfrac{c(\cos\alpha - \cos\beta\cos\gamma)}{\sin\gamma} \\ 0 & 0 & \dfrac{c\sqrt{(1-\cos^2\alpha-\cos^2\beta-\cos^2\gamma+2\cos\alpha\cos\beta\cos\gamma)}}{\sin\gamma} \end{bmatrix}$$

Thus the orthogonal system is defined with X along a, Y in the ab plane, and Z normal to this, as shown in the sketch below:



The symmetry information is supplied as a set of matrices and vectors to transform the original coordinates into symmetric equivalents, according to the equation:

$$\underset{\sim}{x}_i = \underset{\sim}{R}_{ij} . \underset{\sim}{x}_j + \underset{\sim}{V}_i \qquad \text{(Eq. 1)}$$

where x_i = fractional coordinates of generated atoms,

x_j = fractional coordinates of original atoms,

$\underset{\sim}{R}_{ij}$ = symmetry matrix, and

V_i = symmetry vector.

Example: In the space group $P4_1$ (Int. Tables for X-ray Crystallography; Vol.I, p.167), there are 4 equivalent positions, viz., x, y, z; $\bar{x}, \bar{y}$, 1/2+ z; $\bar{y}, x$, 1/4+ z; and $y, \bar{x}$, 3/4+ z.

These positions have the following matrices and vectors:

1) x, y, z

$$\underset{\sim}{R}_{ij} = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \qquad \underset{\sim}{V}_{i} = \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

2) $\bar{x}, \bar{y}, 1/2 + z$

$$\underset{\sim}{R}_{ij} = \begin{bmatrix} -1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \qquad \underset{\sim}{V}_{i} = \begin{bmatrix} 0 \\ 0 \\ 1/2 \end{bmatrix}$$

3) $\bar{y}, x, 1/4 + z$

$$\underset{\sim}{R}_{ij} = \begin{bmatrix} 0 & -1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix} \qquad \underset{\sim}{V}_{i} = \begin{bmatrix} 0 \\ 0 \\ 1/4 \end{bmatrix}$$

4) $y, \bar{x}, 3/4 + z$

$$\underset{\sim}{R}_{ij} = \begin{bmatrix} 0 & 1 & 0 \\ -1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix} \qquad \underset{\sim}{V}_{i} = \begin{bmatrix} 0 \\ 0 \\ 3/4 \end{bmatrix}$$

Centrosymmetric equivalents do not require separate matrices, and certain types of hexagonal and rhombohedral symmetry, as well as I, F, A, B, C-centring, are dealt with separately; so that the maximum number of matrices required is 24.

It often happens, in mineral structures, that many atoms are in special positions and thus do not obey the same symmetry. To cope with this situation, a symmetry list is supplied for each symmetry and the number of the applicable symmetry list is supplied with each atom.

As an example, suppose that a structure with the space group P4 (Int. Tables for X-ray Crystallography Vol. I, p.167) has three independent atoms: the first on position 4d, the second on 2c, and the last on 1a. The matrices for the 4d positions x, y, z; $\bar{x}, \bar{y}, z$; y, $\bar{x}$, z; $\bar{y}$, x, z, respectively, are:

1

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

2

$$\begin{bmatrix} -1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

3

$$\begin{bmatrix} 0 & 1 & 0 \\ -1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

4

$$\begin{bmatrix} 0 & -1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

The 2c positions 0, 1/2, z and 1/2, 0, z may be considered x, y, z and y, x, z, and therefore have matrices 1 and 3 above (0 = -0). The 1a position 0, 0, **z** will, of course, have matrix 1.

There are, therefore, three symmetry lists of matrix numbers:

1 2 3 4

1 3

1

All equivalents within the unit cell are generated for each atom before proceeding to the next atom. The surrounding unit cells are then "scanned" with the total contents of 1 unit cell, all equivalents less than the calculated distance from the centre of the original group being saved. When this procedure has been carried out for all atoms, the complete list of generated atoms is searched for all distances less than the pre-set limit from each of the "original" atoms in turn. The angles between each pair of distances found around each "original" atom are then calculated and printed.

INPUT

The input to the program is via punched cards. In the description, b = blank and x = a number which may be zero or blank. Leading zeros may be left blank. The format of the card description is the same as the format for the first card.

First Card

Card Column (cc)	Contents	Comments
1-3	xxx	Number of structures to be scanned
4-80	Blanks	

The remaining cards are in sets, one set per structure.

Title Card

cc	Contents	Comments
1-70	Title of structure	

Parameter Card

1-10	xxxx.xxxx b	Unit-cell length a (Å)
11-20	xxxx.xxxx b	Unit-cell length b (Å)
21-30	xxxx.xxxx b	Unit-cell length c (Å)
31-40	xxxx.xxxx b	Unit-cell angle α (degrees)
41-50	xxxx.xxxx b	Unit-cell angle β (degrees)
51-60	xxxx.xxxx b	Unit-cell angle γ (degrees)
61-70	xxxx.xxxx b	Radius of scan (Å)
71-74	xxx b	System number (see below)
75-80	xxx bbb	Lattice number (see below)

The system numbers determine the orthogonalization procedure, and have been selected as follows:

Number	System
1	Triclinic
2	Monoclinic
3	Orthorhombic
4	Tetragonal
5	Hexagonal
6	Rhombohedral
7	Cubic

The lattice number is more complex, and is used to control the types of equivalent atoms generated.

Number	Lattice Type
1	P
2	I
3	F
4	A
5	B
6	C
7	Rhombohedral
8	Hexagonal

As an illustration, 4 would produce equivalent atoms at x, y, z and x, 1/2 + y, 1/2 + z; 7 produces atoms at xyz, zxy and yzx; and 8 produces atoms at xyz; 1/3 + x, 2/3 + z; and 2/3 + x, 1/3 X z. If a particular lattice type is specified, all atoms obey this symmetry (as distinct from a centre of symmetry; see Atom Coordinate Cards below). However, in the

case when the system number is 5 or 6, the lattice number need not be 7 or 8; e.g., if the space group is P6, the system number is 5 but the lattice number is 1.

The next cards are the matrices for the symmetry operations, and the number of cells to be scanned in each direction. The program is set to scan three cells in each direction (-1, 0, 1) around the central cell. However, if any of the cell lengths is less than or only slightly greater than the radius of the scan, it is possible that important distances will be missed. In order to avoid this, the number of cells to be scanned can be changed to $2n + 1$ where $n > 1$, in any or all directions; i.e., if five cells are to be scanned in the y direction, $n_y = 2$.

Matrix Cards

The first card is

cc	Content	Comments
1-3	xxx	No. of matrices N (1 per card)
4-6	bxx	n_x } if the normal 3-cell scan is
7-9	bxx	n_y } required, the appropriate
10-12	bxx	n_z } columns may be left blank or punched as 1. If no changes are needed, all columns may be left blank.
13-80	blanks	

There are then N cards, each of which has the format:

cc	Content	Comment	
1-3	xxb	R 11	
4-6	xxb	R 12	
7-9	xxb	R 13	
10-12	xxb	R 21	
13-15	xxb	R 22	1 is b 1 b
16-18	xxb	R 23	-1 is -1 b
19-21	xxb	R 31	
22-24	xxb	R 32	
25-27	xxb	R 33	
28-34	xxxxxxb	V_1	
35-41	xxxxxxb	V_2	
42-48	xxxxxxb	V_3	100,000 x vector; i.e., 1/2 would be 50,000.
49-80	blank		

The next cards are the symmetry list.

Symmetry Cards

cc	Content	Comments
1-3	xxx	No. of lists M (1 per card)
4-80	blank	

which is followed by M cards, each of which has the format:

1-3	xxb	Number of matrices in this list
4-6	xxb	1st matrix number
7-9	xxb	2nd matrix number
.	.	.
.	.	.

cc	Content	Comments
.	.	.
etc.	etc.	etc.
73-75	xxb	24th matrix number
76-80	blank	

Matrix numbers need only be given for the number of matrices specified in the first two columns; the remainder of the card may be left blank.

The remaining cards are atom coordinate cards,

Atom Coordinate Cards

cc	Content	Comments
1-3	xxx	Number of atoms L (1 per card)
4-80	blank	

followed by L cards, each of which contains the following information:

1-5	xxxxb	Atom name, e.g. CUlb		(max. of 4 characters)
6-12	xxxxxxb	x/a	x 100,000	i.e., -0.197 would be -19700
13-19	xxxxxxb	y/b	x 100,000	
20-26	xxxxxxb	z/c	x 100,000	
27-29	b x b	1 if centre of symmetry to be applied to this atom. 0 if centre of symmetry not to be applied to this atom.		
30-32	x x b	Number of symmetry list to be applied to this atom.		
33-80	blank			

For subsequent structures, insert the appropriate title card and the cards relating to it. In order to run the program on the IBM 360 computer, the last card of the last structure should be followed by a card with /* in card columns 1 and 2 and the remainder blank.

OUTPUT

The output is fairly self-explanatory. The first page is headed by the program name, followed by the structure title, the cell parameters and the orthogonality matrix, in the form:

$$\begin{bmatrix} T_{11} & T_{12} & T_{13} \\ 0 & T_{22} & T_{23} \\ 0 & 0 & T_{33} \end{bmatrix}$$

The atomic coordinates come next, with the atom number (L) and the orthogonal coordinates in Å. The last four numbers on the page are the orthogonal dimensions of the block defined by the input atoms, and the total radius of the circumscribing sphere, all in Å.

The remaining pages need hardly any explanation. On the first pages are listed the equivalents. All saved equivalents are presented on separate pages for each atom. On each line, the first number is 1000L plus the number of this equivalent; the next three numbers are the coordinates in fractions of a unit cell; next is the number of this equivalent in the equivalent list of all atoms; and the last three numbers are the orthogonalized coordinates in Å.

The subsequent pages give the bonds and angles around each atom. Before each list of angles, 1000L and the atom name are printed for all atoms, to facilitate interpretation of the lists of lengths and angles. In the following lines, the number in the first column gives the atom 1000L plus equivalent

number 1; that in the second column gives the name of the central atom; and that in the third gives the atom 1000L plus equivalent number 2. The fourth and fifth columns give the distances (Å) between the first and central atoms and between the third and central atoms, respectively. The final number in the line is the angle centred on the central atom. Subsequent lines give the lengths and angles from the central atom to other neighbouring ones, e.g.,

3006 OXY1 1017 d_1 d_2 Θ,

in which d_1 and d_2 are the two bond lengths, and Θ is the angle centred on OXY1 to the 6th equivalent of atom 3 and the 17th equivalent of atom 1.

= = = =

EJG:(PES):vb

PROGRAM LISTING

FORTRAN IV G LEVEL 0, MOD 0

```
C  BOND SCAN PROGRAM
      DIMENSION U(4,2500),A(3),ANG(3),CS(3),T(3,3),R(24,3,3),V(24,3),
     1NUM(16),NM(16,24),X(99),Y(99),Z(99),IX(96),IY(96),IZ(96),
     2NAME(99),DIS(4,400),IDIS(400),ICENT(99),ISYM(99)
      INTEGER R,X,Y,Z,U,V
      LNPT=3
      LNCD=1
      SAD=3.14159/180.0
      READ(LNCD,5)NSTRUC
      DO 170 NTYPE=1,NSTRUC
      WRITE(LNPT,1)
    1 FORMAT(1H1,///30X,19HBOND AND ANGLE SCAN,//)
      READ(LNCD,2)
      WRITE(LNPT,2)
    2 FORMAT(50H
     1                )
      READ(LNCD,3)A(1),A(2),A(3),ANG(1),ANG(2),ANG(3),RAD,ISYS,LAT
      WRITE(LNPT,4)A(1),A(2),A(3),ANG(1),ANG(2),ANG(3),RAD,ISYS,LAT
    3 FORMAT(7(F9.4,1X),2(I3,1X))
    4 FORMAT(/12X,7(F9.4,1X),2(I3,1X),/)
      DO 99 I=1,3
      DO 99 J=1,3
   99 T(I,J)=0.0
C  COMPUTE ORTHOGONALITY MATRIX
      GO TO(100,110,120,120,910,100,120),ISYS
  100 DO 101 I=1,3
  101 CS(I)=COS(ANG(I)*SAD)
      SN=SIN(ANG(3)*SAD)
      T(1,1)=A(1)
      T(1,2)=A(2)*CS(3)
      T(1,3)=A(3)*CS(2)
      T(2,2)=A(2)*SN
      T(2,3)=(A(3)*(CS(1)-CS(2)*CS(3)))/SN
      T(3,3)=(A(3)/SN)*SQRT(1.0-CS(1)**2-CS(2)**2-CS(3)**2+2.0*CS(1)*
     1CS(2)*CS(3))
      GO TO 122
  110 CS(2)=COS(ANG(2)*SAD)
      SN=SIN(ANG(2)*SAD)
      T(1,1)=A(1)
      T(1,3)=A(3)*CS(2)
      T(2,2)=A(2)
      T(3,3)=A(3)*SN
      GO TO 122
  910 T(1,1)=A(1)
      T(1,2)=A(2)*COS(ANG(3)*SAD)
      T(2,2)=A(2)*SIN(ANG(3)*SAD)
      T(3,3)=A(3)
      GO TO 122
```

```
  120  DO 121 I=1,3
  121  T(I,I)=A(I)
C      READ MATRICES FOR EQUIVALENT POSITIONS
  122  WRITE(LNPT,19)
  19   FORMAT(/20X,20HORTHOGONALITY MATRIX)
       DO 51 I=1,3
  51   WRITE(LNPT,50)T(I,1),T(I,2),T(I,3)
  50   FORMAT(/20X,3(F10.6,3X))
       WRITE(LNPT,18)
  18   FORMAT(/)
       READ(LNCD,20)NMAT,NX,NY,NZ
  20   FORMAT(I3,3(1X,I2))
  5    FORMAT(I3)
       IF(NX.EQ.0) NX=1
       KSTEP=100000*NX+0.5
       KLIM=NX+NX+1
       KRANGE=100000*KLIM+0.5
       IF(NY.EQ.0) NY=1
       LSTEP=100000*NY+0.5
       LLIM=NY+NY+1
       LRANGE=100000*LLIM+0.5
       IF(NZ.EQ.0) NZ=1
       MSTEP=100000*NZ+0.5
       MLIM=NZ+NZ+1
       MRANGE=100000*MLIM+0.5
       DO 123 I=1,NMAT
  123  READ(LNCD,6)R(I,1,1),R(I,1,2),R(I,1,3),R(I,2,1),R(I,2,2),
      1R(I,2,3),R(I,3,1),R(I,3,2),R(I,3,3),V(I,1),V(I,2),V(I,3)
  6    FORMAT(9(I2,1X),3(I6,1X))
C  READ TYPES OF SYMMETRY
       READ(LNCD,5)NSYM
       DO 124 I=1,NSYM
  124  READ(LNCD,7)NUM(I),NM(I,1),NM(I,2),NM(I,3),NM(I,4),NM(I,5),
      1NM(I,6),NM(I,7),NM(I,8),NM(I,9),NM(I,10),NM(I,11),NM(I,12),
      2NM(I,13),NM(I,14),NM(I,15),NM(I,16),NM(I,17),NM(I,18),NM(I,19),
      3NM(I,20),NM(I,21),NM(I,22),NM(I,23),NM(I,24)
  7    FORMAT(25(I2,1X))
C      READ COORDINATE CARDS AND FIND C OF G AND RADIUS
       READ(LNCD,5)NATOM
       DO 125 I=1,NATOM
       READ(LNCD,8)NAME(I),X(I),Y(I),Z(I),ICENT(I),ISYM(I)
       AX=(X(I)*T(1,1)+Y(I)*T(1,2)+Z(I)*T(1,3))*0.00001
       AY=(Y(I)*T(2,2)+Z(I)*T(2,3))*0.00001
       AZ=Z(I)*T(3,3)*0.00001
  125  WRITE(LNPT,9)NAME(I),X(I),Y(I),Z(I),ICENT(I),ISYM(I),I,AX,AY,AZ
  8    FORMAT(A4,3(1X,I6),2(1X,I2))
  9    FORMAT(/9X,A4,3(4X,I6),2(1X,I2),3X,I3,3(3X,F8.4))
       IXMIN=X(1)
```

```
      IXMAX=X(1)
      IYMIN=Y(1)
      IYMAX=Y(1)
      IZMIN=Z(1)
      IZMAX=Z(1)
      DO 137 I=2,NATOM
      IF(X(I).LT.IXMIN) IXMIN=X(I)
      IF(X(I).GT.IXMAX) IXMAX=X(I)
      IF(Y(I).LT.IYMIN) IYMIN=Y(I)
      IF(Y(I).GT.IYMAX) IYMAX=Y(I)
      IF(Z(I).LT.IZMIN) IZMIN=Z(I)
      IF(Z(I).GT.IZMAX) IZMAX=Z(I)
  137 CONTINUE
      CX=(IXMAX+IXMIN)*0.000005
      CY=(IYMAX+IYMIN)*0.000005
      CZ=(IZMAX+IZMIN)*0.000005
      CX=CX*T(1,1)+CY*T(1,2)+CZ*T(1,3)
      CY=CY*T(2,2)+CZ*T(2,3)
      CZ=CZ*T(3,3)
      DELX=IXMAX-IXMIN
      DELY=IYMAX-IYMIN
      DELZ=IZMAX-IZMIN
      D=SQRT((DELX*T(1,1)+DELY*T(1,2)+DELZ*T(1,3))**2+
     1(DELY*T(2,2)+DELZ*T(2,3))**2+(DELZ*T(3,3))**2)
      D=D*0.000005+RAD
      WRITE(LNPT,52)CX,CY,CZ,D
   52 FORMAT(//20X,4(F10.6,3X))
      D=D*D
C  FIND ALL EQUIVALENTS OF EACH ATOM
      ITOT=1
      DO 150 I=1,NATOM
      N=ISYM(I)
      M=NUM(N)
      DO 151 J=1,M
      L=NM(N,J)
      IX(J)=R(L,1,1)*X(I)+R(L,1,2)*Y(I)+R(L,1,3)*Z(I)+V(L,1)
      IY(J)=R(L,2,1)*X(I)+R(L,2,2)*Y(I)+R(L,2,3)*Z(I)+V(L,2)
  151 IZ(J)=R(L,3,1)*X(I)+R(L,3,2)*Y(I)+R(L,3,3)*Z(I)+V(L,3)
      GO TO(410,420,430,440,450,460,470,480),LAT
  420 DO 421 J=1,M
      IX(J+M)=IX(J)+50000
      IY(J+M)=IY(J)+50000
  421 IZ(J+M)=IZ(J)+50000
      NO=M+M
      GO TO 411
  430 DO 431 J=1,M
      IX(J+M)=IX(J)
      IX(J+2*M)=IX(J)+50000
```

```
      IX(J+3*M)=IX(J)+50000
      IY(J+M)=IY(J)+50000
      IY(J+2*M)=IY(J)
      IY(J+3*M)=IY(J)+50000
      IZ(J+M)=IZ(J)+50000
      IZ(J+2*M)=IZ(J)+50000
  431 IZ(J+3*M)=IZ(J)
      NO=4*M
      GO TO 411
  440 DO 441 J=1,M
      IX(J+M)=IX(J)
      IY(J+M)=IY(J)+50000
  441 IZ(J+M)=IZ(J)+50000
      NO=M+M
      GO TO 411
  450 DO 451 J=1,M
      IX(J+M)=IX(J)+50000
      IY(J+M)=IY(J)
  451 IZ(J+M)=IZ(J)+50000
      NO=M+M
      GO TO 411
  460 DO 461 J=1,M
      IX(J+M)=IX(J)+50000
      IY(J+M)=IY(J)+50000
  461 IZ(J+M)=IZ(J)
      NO=M+M
      GO TO 411
  470 DO 471 J=1,M
      IX(J+M)=IZ(J)
      IX(J+2*M)=IY(J)
      IY(J+M)=IX(J)
      IY(J+2*M)=IZ(J)
      IZ(J+M)=IY(J)
  471 IZ(J+2*M)=IX(J)
      NO=3*M
      GO TO 411
  480 DO 481 J=1,M
      IX(J+M)=IX(J)+33333
      IY(J+M)=IY(J)+66667
      IZ(J+M)=IZ(J)+66667
      IX(J+2*M)=IX(J)+66667
      IY(J+2*M)=IY(J)+33333
  481 IZ(J+2*M)=IZ(J)+33333
      NO=3*M
      GO TO 411
  410 NO=M
  411 IF(ICENT(I).EQ.0) GO TO 414
      DO 413 J=1,NO
```

```
      IX(J+NO)=-IX(J)
      IY(J+NO)=-IY(J)
  413 IZ(J+NO)=-IZ(J)
      NO=NO+NO
C  FIND ALL ATOMS SURROUNDING WITHIN SPHERE D
  414 IDENT=1000*I+1
      WRITE(LNPT,10)NAME(I)
   10 FORMAT(1H1,//15X,20HEQUIVALENTS OF ATOM ,A4,13HWITHIN SPHERE,/)
      DO 152 J=1,NO
      IX(J)=IX(J)-KSTEP
      IY(J)=IY(J)-LSTEP
  152 IZ(J)=IZ(J)-MSTEP
      DO 150 J=1,NO
      DO 150 K=1,MLIM
      DO 154 L=1,LLIM
      DO 153 M=1,KLIM
      GO TO(500,510,520,520,530,500,520),ISYS
  500 AX=(IX(J)*T(1,1)+IY(J)*T(1,2)+IZ(J)*T(1,3))*0.00001
      AY=(IY(J)*T(2,2)+IZ(J)*T(2,3))*0.00001
      AZ=IZ(J)*T(3,3)*0.00001
      GO TO 160
  510 AX=(IX(J)*T(1,1)+IZ(J)*T(1,3))*0.00001
      GO TO 540
  520 AX=IX(J)*T(1,1)*0.00001
      GO TO 540
  530 AX=(IX(J)*T(1,1)+IY(J)*T(1,2))*0.00001
  540 AY=IY(J)*T(2,2)*0.00001
      AZ=IZ(J)*T(3,3)*0.00001
  160 DIST=(AX-CX)*(AX-CX)+(AY-CY)*(AY-CY)+(AZ-CZ)*(AZ-CZ)
      IF(D.LT.DIST) GO TO 153
      U(1,ITOT)=IDENT
      U(2,ITOT)=100000*AX
      U(3,ITOT)=100000*AY
      U(4,ITOT)=100000*AZ
      WRITE(LNPT,11)IDENT,IX(J),IY(J),IZ(J),ITOT,AX,AY,AZ
   11 FORMAT(5X,I6,6X,3(I7,3X),3X,I4,3X,3(3X,F8.4))
      IDENT=IDENT+1
      ITOT=ITOT+1
  153 IX(J)=IX(J)+100000
      IX(J)=IX(J)-KRANGE
  154 IY(J)=IY(J)+100000
      IY(J)=IY(J)-LRANGE
  150 IZ(J)=IZ(J)+100000
      DO 155 N=1,NATOM
      X(N)=X(N)*T(1,1)+Y(N)*T(1,2)+Z(N)*T(1,3)
      Y(N)=Y(N)*T(2,2)+Z(N)*T(2,3)
  155 Z(N)=Z(N)*T(3,3)
      RAD=RAD*RAD
```

```
      ITOT=ITOT-1
C  FIND AND STORE ALL BONDS BETWEEN ORIGINAL AND SAVED ATOMS
      DO 170 N=1,NATOM
      NBOND=0
      WRITE(LNPT,14)NAME(N)
   14 FORMAT(1H1,//15X,26HBONDS AND ANGLES FOR ATOM ,A4)
      DO 169 NN=1,NATOM
      NTHOU=1000*NN
  169 WRITE(LNPT,17)NTHOU,NAME(NN)
   17 FORMAT(/15X,I6,6X,A4)
      DO 172 ITAT=1,ITOT
      DX=(U(2,ITAT)-X(N))*0.00001
      DY=(U(3,ITAT)-Y(N))*0.00001
      DZ=(U(4,ITAT)-Z(N))*0.00001
      BOND=DX*DX+DY*DY+DZ*DZ
      IF(RAD.LT.BOND.OR.BOND.LT.0.01) GO TO 172
      NBOND=NBOND+1
      DIS(1,NBOND)=DX
      DIS(2,NBOND)=DY
      DIS(3,NBOND)=DZ
      DIS(4,NBOND)=SQRT(BOND)
      IDIS(NBOND)=ITAT
  172 CONTINUE
C  FIND ALL ANGLES BETWEEN BONDS
      WRITE(LNPT,15)NBOND
   15 FORMAT(/15X,10HTHERE ARE ,I3,6H BONDS)
      NANG=0.5*NBOND*(NBOND-1)+0.5
      WRITE(LNPT,16)NANG
   16 FORMAT(/15X,10HTHERE ARE ,I3,7H ANGLES,/)
      IF(NBOND.GT.1) GO TO 173
      IF(NBOND.GT.0) GO TO 174
      WRITE(LNPT,60)
   60 FORMAT(/10X,18HTHERE ARE NO BONDS)
      GO TO 170
  174 NOM=IDIS(1)
      WRITE(LNPT,61)U(1,NOM),DIS(4,1)
   61 FORMAT(/6X,I6,6X,F8.4)
      GO TO 170
  173 MAX1=NBOND-1
      DO 170 N1=1,MAX1
      MIN2=N1+1
      DO 170 N2=MIN2,NBOND
      CANG=(DIS(1,N1)*DIS(1,N2)+DIS(2,N1)*DIS(2,N2)+DIS(3,N1)*DIS(3,N2)
     1/(DIS(4,N1)*DIS(4,N2))
      IF(CANG.GT.1.0) CANG=1.0
      IF(CANG.LT.-1.0) CANG=-1.0
      THETA=57.29578*ARCOS(CANG)
      NOM=IDIS(N1)
```

```
      I1=U(1,NOM)
      NOM=IDIS(N2)
      I2=U(1,NOM)
      WRITE(LNPT,12)I1,NAME(N),I2,DIS(4,N1),DIS(4,N2),THETA
170   CONTINUE
12    FORMAT(6X,I6,6X,A4,4X,I6,3(6X,F8.3))
      STOP
      END
```